
ABSTRACT 

Can you create hundreds of great looking excel tables all within SAS® and make them all Section 508 compliant at 

the same time?  This paper will examine how to use ODS TAGSETS EXCELXP and other BASE SAS® features to 

create fantastic looking excel worksheet tables that are all Section 508 compliant.   This paper will demonstrate that 

there is no need for any outside intervention or pre or post-meddling with the excel files to make them Section 508 

compliant.  We do all with simple BASE SAS® code.   

INTRODUCTION 

Section 508 requires that federal agencies make online information and services fully available to the approximately 

54 million Americans who have vision disabilities, many of whom cannot access information that is not Section 508 

compliant.  Those with vision disabilities use assisted devices called screen readers to read the contents of the 

screen or website to them.  This paper shows how the U.S. Bureau of the Census created hundreds of Excel tables 

to be published on its website for its 2012 Annual Social and Economic Supplement (ASEC) to the Current 

Population Survey (CPS) Poverty report and made all of them Section 508 compliant.  The excel tables can be found 

at  http://www.census.gov/hhes/www/cpstables/032012/pov/toc.htm 

It will show that no pre or post-processing was done to the excel files.  No empty excel shells were created first and 

made Section 508 compliant and later filled with data via Dynamic Data Exchange (DDE) or other post-processing 

methods.  Rather, we create hundreds of section 508 Excel files using simple BASE SAS® code.  We examine which 

parts of any Excel table need to be made 508 compliant, including cell A1, rows with indentation, and footnote links.   

Moreover, we review how we create the excel files with ODS TAGSETS.EXCELXP.  Additionally, we reveal a few 

ODS tricks with style in PROC FORMAT, title and footnote statements to make the Excel tables Section 508 

compliant. 

WHAT IS SECTION 508 COMPLIANCE? 

Section 508 requires that federal agencies make online information and services fully available to the approximately 

54 million Americans who have vision disabilities, many of whom cannot access information that is not Section 508 

compliant.   For HTML or Excel files, the data must be readable for both those with or without disabilities.  This 

sounds difficult at first with possible pre or post-processing to accomplish it.   However, after closer examination, it is 

not nearly as difficult and all is possible within BASE SAS® code.  

SAS® Code to Make Excel Files Section 508 Compliant 

The first thing that we need to do in our SAS® code to make our Excel file Section 508 compliant is to insert a new 

row at cell A1 of the Excel file.  This new row in cell A1 needs to contain a description of the Excel table listing the 

location of all column and row headers for all sub-tables in the Excel file.  We can do this by using a title1 statement 

in our SAS® program.  However, we need to make this row invisible or unseen by those without a vision disability.  

To do this we can change the color of the text in our title1 statement to white on a white background. 

title1 j=left color=white "Table with row headers in column A and column headers in 

rows 9 through 10, 26 through 27, 43 through 44 and 60 through 61”; 

Display 1.  Excel Table with a Table Description in cell A! 

Cell A1 is hidden in Display 1.  However, when placing the cursor on cell A1, the contents are revealed in the display 
box at the top of the screen under the “Page Layout” pull-down menu and above column B.  The screen reader will 
pick up this title in cell A1 with the description of the table.  For those with disabilities, they will know where the row 

and column headers of each sub-table are located in the Excel file.  For those without disabilities, cell A1 will go 

unnoticed because we changed the color to white.  Essentially, the color is “white on white” and cell A1 appears 
invisible to the naked eye. 
 
Next, we need to alter all rows that are indented.  In each of our ASEC Poverty tables, we have several rows that are 
indented.  Screen readers cannot read indents or blank spaces.  Thus, we need to add “white dots” to each row 
header that has indentation.  We will have to add one white dot for each space or indent. 
 

proc format;                                                                                                                             

value agef (notsorted multilabel)                                                                                                        

  0-17    = '~S={foreground=white}..~S={}Under 18 years'                                                                         

  0-4     = '~S={foreground=white}....~S={}Under 5 years (1)'                                                                            

  5-17    = '~S={foreground=white}....~S={}5 to 17 years'                                                                                

  18-64   = '~S={foreground=white}..~S={}18 to 64 years'                                                                                 

  18-24   = '~S={foreground=white}....~S={}18 to 24 years'                                                                               

  25-34   = '~S={foreground=white}....~S={}25 to 34 years'                                                                               

  35-44   = '~S={foreground=white}....~S={}35 to 44 years'                                                                               

  45-54   = '~S={foreground=white}....~S={}45 to 54 years'                                                                               

  55-59   = '~S={foreground=white}....~S={}55 to 59 years'                                                                               

  60-64   = '~S={foreground=white}....~S={}60 to 64 years'                                                                               

  65-high = '~S={foreground=white}..~S={}65 years and over'                                                                      

  65-74   = '~S={foreground=white}....~S={}65 to 74 years'                                                                               

  75-high = '~S={foreground=white}....~S={}75 years and over'                                                                            

        ;                                                                                                                                

run;   

         

Notice the use of style elements in the PROC FORMAT.  We change the color of the format to white by using the 
foreground style and changing its value to white.  We than add periods or dots for each space of indentation that is 
necessary for that row header value.  We then close out the style and thus, change the foreground color back to 
black.  Black is the default foreground color.  Thus, only the dots are color white.  The rest of the row header is black.  
Thus, the white dots are unseen by those without disabilities.  They appear as blank spaces to the naked eye.  
However, for those with disabilities the screen reader picks up the dots.  The screen reader alerts the user to the 
indentation.  The user with the vision disability knows that this is a row header with indentation.  Display 2 shows no 
dots or periods in  cell A13.  However, we see the “white dots” in the display box at the top of the screen below the 
“Page Layout” pull-down menu and above column B.  
 
 

 

Display 2.  Excel Table with “White Dots” for the rows with indentation 

 
Our ASEC Poverty tables have footnote links at the bottom of each table.  We need to be identify and label these for 
the screen reader.   To identify the footnote link, we add a row above the link to alert the screen reader to the footnote 
link in the next row.  The description in this row needs to identify the row number and column header of the footnote 
link.                                                                                                                                      
                                                                                                                                         

%if &i=4 %then %do;  /* There are 4 sub-tables in the Excel worksheet */                                                                 

footnote1 j=l color=white h=9pt "A website link to the footnote descriptions is in row 

75, column A";                                    

footnote2 j=l h=9pt underlin=1 

link="http://www.census.gov/hhes/www/cpstables/03&marchcps./povnotes.htm" "Poverty 

Footnotes";            

%end;           

In the code above, the footnote1 statement contains the description and location of the footnote link.  Notice that we 
change the color of the text to white, so that it is unseen by the naked eye.  However, the screen reader will read this 
text and alert the user with a vision disability to a footnote link in row 75, column A of the table.  The footnote2 
statement contains the actual footnote link with the appropriate URL address. 

 
Hundreds of Section 508 Excel Files 

We exploit ODS TAGSETS EXCELXP to create the excel files.  We wrap the ODS TAGSETS EXCELXP  in a macro 

loop to create multiple excel files.  We utilize several ExcelXP Tagset options to get the table to look exactly the way 

we want.  Our Excel worksheet has several sub-tables.  For Section 508 Compliance, there should not be a blank 

space between the sub-tables.  To accomplish this we add the skip_space option to the ODS TAGSETS EXCELXP 

statement.  This option allows you to specify spacing between objects in a report with the objects listed in a comma 

separated list.  The first value in the list for skip_space corresponds to the spacing between the sub-tables.  We want 

no spaces between sub-tables, so the first value in the list for skip_space is “0”.                                                                                                                                                                                                                                                   

                                                                                                                                                                                                                                                          

%macro iterations;  /* Create 135 Excel tables by race and poverty group */                                                                                                                                                                               

%do raceit=1 %to 9;  /* Nine Race Groups */                                                                                                                                                                                                       

     %do percentit=1 %to 15;  /* Fifteen Poverty Group Levels */                                                                                                                                                                                          

                                                                                                                                                                                                                                                          

%macro print_report_excel;                                                                                                                                                                                                                                

ods escapechar='~';                                                                                                                                                                                                                                       

ods tagsets.excelxp path="&outp" file="POV01_&ratio2._&raceit..xls"                                                                                                                                                                                       

style=styles.newstyle options(header_data_associations="yes" autofit_height='yes')                                                                                                                                                                        

options(embedded_titles="yes" embedded_footnotes='yes' sheet_interval='none' 

absolute_column_width="24,10,10,10,10,10,10,10,10,10,10,10,10"                                                                                                               

sheet_name='POV01' skip_space="0,0,0,1,1,0,0");                                                                                                                                                                                                           

                                                                                                                                                                                                                                                          

title1 j=left color=white "Table with row headers in column A and column headers in 

rows 9 through 10, 26 through 27, 43 through 44, and 60 through 61";                                                                                                  

title2 j=left bold height=11pt "POV01: Age and Sex of All People, Family Members and 

Unrelated Individuals Iterated by Income-to-Poverty Ratio and Race: &incyear";                                                                                       

title3 j=left bold height=11pt "Below &ratio2.% of Poverty -- &race4";                                                                                                                                                                                    

title4 j=left height=9pt underlin=1 

link="http://www.census.gov/apsd/techdoc/cps/cpsmar&yr..pdf" "For information on 

confidentiality protection, sampling error, nonsampling error, and definitions, see 

www.census.gov/apsd/techdoc/cps/cpsmar&yr..pdf"; 

title5 j=left height=9pt "Source: U.S. Census Bureau, Current Population Survey, 

&marchcps Annual Social and Economic Supplement.";                                                                                                                       

title6 j=left height=9pt "(Numbers in thousands)";                                                                                                                                                                                                        

                                                                                                                                                                                                                                                          

                                                                                                                                                                                                                                                          

%do i=1 %to 4; /* There are 4 sub-tables in the Excel worksheet */                                                                                                                                                                                        

                                                                                                                                                                                                                                                          

/* generate output */                                                                                                                                                                                                                                     

                                                                                                                                                                                                                                                          

%if &i=1 %then %do;                                                                                                                                                                                                                                       

 title7 " ";                                                                                                                                                                                                                                              

 title8 j=left bold height=9pt "&&title&i";                                                                                                                                                                                                               

%end;                                                                                                                                                                                                                                                     

%else %do;                                                                                                                                                                                                                                                

 title1;title2;title3;title4;title5;title6;title7;title8;                                                                                                                                                                                                 

  title1  j=left bold height=9pt "&&title&i";                                                                                                                                                                                                             

%end;                                                                                                                                                                                                                                                     

                                                                                                                                                                                                                                                          

proc report data=section&i nowd spanrows split='|' out=rept&i;                                                                                                                                                                                            

column rowhead2 fam_type, (count_sum poverty_sum poverty_pct);                                                                                                                                                                                            

define rowhead2 / "&race2 | Below &ratio2.%" group order=data flow style(column) = 

[protectspecialchars=off];                                                                                                                                             

define fam_type / '' across order=data preloadfmt missing ;                                                                                                                                                                                               

define count_sum / 'All | income | levels' style(column)=[tagattr="format:#,##0"];                                                                                                                                                                        

define poverty_sum / "Below | &ratio2.% of | poverty (5)" 

style(column)=[tagattr="format:#,##0"];                                                                                                                                                         

define poverty_pct / "Percentage | below &ratio2.% of | poverty | (6)"                                                                                                                                                                                    

f=pctf. style(column)=[tagattr="format:##0.0"] ;                                                                                                                                                                                                          

run;                                                                                                                                                                                                                                                      

%end;                                                                                                                                                                                                                                                     

%print_report_excel;                                                                                                                                                                                                                                      

footnote;                                                                                                                                                                                                                                                 

ods _all_ close;                                                                                                                                                                                                                                          

                                                                                                                                                                                                                                                          

  %end;                                                                                                                                                                                                                                                   

 %end;                                                                                                                                                                                                                                                    

%mend iterations;                                                                                                                                                                                                                                         

%iterations;             

CONCLUSION 

Yes, it is possible to create hundreds of Section 508 Compliant Excel files using BASE SAS® code without any post 

or pre-processing.  We showed how to add a table description in cell A1 of the Excel file.  Moreover, we revealed how 

to create “white dots” for indented row headers using PROC FORMAT with style to make the dots white and invisible 

to the naked eye.  Furthermore, we discussed adding a row to label the upcoming footnote link to alert the user with a 

vision disability.  Lastly, we exploited ODS TAGSETS EXCELXP to create multiple excel files and made use of the 

skip_space option to assure no spacing between sub-tables.   Creating hundreds of Section 508 Excel files need not 

be a tedious task.  You do not need to process the Excel file outside of SAS® either before or after the fact.  You can 

create hundreds of  fantastic looking Section 508 Compliant Excel tables all within BASE SAS® code. 

